INTRODUCTION

This document supplements the PSpice models SV811-3.INC and SV811-10.INC, and provides some background operation to the operation of the model along with details of functionality modelled or not modelled as the case may be.

Whilst every care has been taken to duplicate the functionality of the modelled device as described here, it should be stressed that modelling is not a substitution for breadboarding or other prototyping methods.

No warranty of any kind is provided for this model, and no liability is assumed for any damage or loss arising out of the use of this model, or application of the results of this model. All trademarks acknowledged. The model is copyright ©1997-2003 Duncan Amplification, and is made available for educational or non-profit use.

MODELLED FUNCTIONS

Inter-electrode capacitance, cathode saturation, grid current.

FUNCTIONS NOT MODELLED

A heater model is not implemented at this stage.
MODEL PERFORMANCE

Figure 1 above shows anode current against swept anode voltage for a range of grid voltages between -110V and +70V in steps of 10V.

MODEL DESCRIPTION

The following describes the various components of the model and their interaction:

Eat  
Is the arctangent curve which is used to calculate the rise in grid current at low values of Va.

Eme  
is the cathode emission limit.

Egs  
is the emission contribution from the grid and anode.

Egs2  
is Egs after raising to the power of 3/2 and factored by a constant so that it may be turned directly into a current value.

Ecath  
is the cathode current value. This is the current between anode and cathode.

Ga  
is the actual cathode current. Synonymous with Ecath.

Gg  
is the grid current.
ALTERING THE MODEL FOR OTHER SIMULATORS

A Spice 3f4 compatible model is available for other simulators. These are designated SV811-3s.INC and SV811-10s.INC.

Duncan Munro
16th June 1997

Web page: http://www.duncanamps.com/
Forum: http://forum.duncanamps.com/
Email: postmaster@duncanamps.com
MODEL LISTING SV811-3

*-----------------------------------------------------------------------
* Filename:  SV811-3.inc  V2  16/6/97
* Simulator:  PSpice
* Device type:  Power triode
* Device model:  Svetlana SV811-3
* Author:  Duncan Munro
* Date:   14/6/97
* Copyright:  (C)1997-2003 Duncan Amplification
*
* The following parameters are not modelled:
* (1) Filament: A "cathode" connection is made which
* should be connected to a potential which is the
* average of the two filament terminals.
* Please note that this model is provided "as is" and
* no warranty is provided in respect of its suitability
* for any application.
* This model is provided for educational and non-profit use.
* Email queries to postmaster@duncanamps.com
*
* Pins   A  Anode
*        G  Grid
*        K  Cathode
*
*-----------------------------------------------------------------------

.SUBCKT SV811-3 A G K
*
* Calculate contribution to cathode current
*
Eat at 0 VALUE={0.636*ATAN(V(A,K)/30)}
Eme me 0 VALUE={PWR(LIMIT(V(A,K),0,10000),1.5)/4500}
Emu mu 0 VALUE={V(G,K)}
Egs gs 0 VALUE={LIMIT(V(A,K)+V(mu)*3.4,0,1E6)}
Egs2 gs2 0 VALUE={PWR(V(gs),1.5)*25E-6}
Ecath cc 0 VALUE={LIMIT(V(gs2),0,V(me))}
*
* Calculate anode current
*
Ga      A      K      VALUE={V(cc)}
*
* Calculate grid current
*
Gg G K VALUE={PWR(LIMIT(V(G,K)+1,0,1E6),1.5)*(2-V(at))*30E-6}
*
* Capacitances
*
Cgk G K 7p
Cga A G 8p
*.ENDS
MODEL LISTING SV811-10

*-----------------------------------------------------------------------
* Filename:  SV811-10.inc  V2  16/6/97
* Simulator:  PSpice
* Device type:  Power triode
* Device model:  Svetlana SV811-10
* Author:  Duncan Munro
* Date:   14/6/97
* Copyright:  (C)1997-2003 Duncan Amplification
*
* The following parameters are not modelled:
* 
* (1) Filament: A "cathode" connection is made which
* should be connected to a potential which is the
* average of the two filament terminals.
* 
* Please note that this model is provided "as is" and
* no warranty is provided in respect of its suitability
* for any application.
* 
* This model is provided for educational and non-profit use.
*
* Email queries to postmaster@duncanamps.com
* 
* Pins   A  Anode
*        G  Grid
*        K  Cathode
*
*-----------------------------------------------------------------------

.SUBCKT SV811-10 A G K
*
* Calculate contribution to cathode current
*
Eat  at  0  VALUE={0.636*ATAN(V(A,K)/30))}
Eme  me  0  VALUE={PWR(LIMIT{V(A,K),0,10000},1.5)/1800}
Emu  mu  0  VALUE={V(G,K)}
Egs  gs  0  VALUE={LIMIT{V(A,K)+V(mu)*8.7,0,1E6}}
Egs2 gs2  0  VALUE={PWR(V(gs),1.5)*12E-6}
Ecath cc  0  VALUE={LIMIT{V(gs2),0,V(me)}}
*
* Calculate anode current
*
Ga  A  K  VALUE={V(cc)}
*
* Calculate grid current
*
Gg  G  K  VALUE={PWR(LIMIT{V(G,K)+1,0,1E6},1.5)*(2-V(at))*73E-6}
*
* Capacitances
*
Cgk  G  K  7p
Cga  A  G  8p
.
. ENDS